

Concurrent Refinement Algebra and Rely Quotients

Julian Fell and Ian Hayes and Andrius Velykis

February 6, 2026

Abstract

The concurrent refinement algebra developed here is designed to provide a foundation for rely/guarantee reasoning about concurrent programs. The algebra builds on a complete lattice of commands by providing sequential composition, parallel composition and a novel weak conjunction operator. The weak conjunction operator coincides with the lattice supremum providing its arguments are non-aborting, but aborts if either of its arguments do. Weak conjunction provides an abstract version of a guarantee condition as a guarantee process. We distinguish between models that distribute sequential composition over non-deterministic choice from the left (referred to as being conjunctive in the refinement calculus literature) and those that don't. Least and greatest fixed points of monotone functions are provided to allow recursion and iteration operators to be added to the language. Additional iteration laws are available for conjunctive models. The rely quotient of processes c and i is the process that, if executed in parallel with i implements c . It represents an abstract version of a rely condition generalised to a process.

Contents

1	Overview	4
2	Refinement Lattice	4
3	Sequential Operator	6
3.1	Basic sequential	6
3.2	Distributed sequential	7
4	Parallel Operator	9
4.1	Basic parallel operator	9
4.2	Distributed parallel	9
5	Weak Conjunction Operator	10
5.1	Distributed weak conjunction	11
6	Concurrent Refinement Algebra	12
7	Galois Connections and Fusion Theorems	14
7.1	Lower Galois connections	15
7.2	Greatest fixpoint fusion theorems	15
7.3	Upper Galois connections	16
7.4	Least fixpoint fusion theorems	16
8	Iteration	17
8.1	Possibly infinite iteration	17
8.2	Finite iteration	18
8.3	Infinite iteration	19
8.4	Combined iteration	20
9	Sequential composition for conjunctive models	21
10	Infimum nat lemmas	22
11	Iteration for conjunctive models	23
12	Rely Quotient Operator	24
12.1	Basic rely quotient	25
12.2	Distributed rely quotient	26
13	Conclusions	28

1 Overview

The theories provided here were developed in order to provide support for rely/guarantee concurrency [6, 5]. The theories provide a quite general concurrent refinement algebra that builds on a complete lattice of commands by adding sequential and parallel composition operators as well as recursion. A novel weak conjunction operator is also added as this allows one to build more general specifications. The theories are based on the paper by Hayes [3], however there are some differences that have been introduced to correct and simplify the algebra and make it more widely applicable. See the appendix for a summary of the differences.

The basis of the algebra is a complete lattice of commands (Section 2). Sections 3, 4 and 5 develop laws for sequential composition, parallel composition and weak conjunction, respectively, based on the refinement lattice. Section 6 brings the above theories together. Section 7 adds least and greatest fixed points and there associated laws, which allows finite, possibly infinite and strictly infinite iteration operators to be defined in Section 8 in terms of fixed points.

The above theories do not assume that sequential composition is conjunctive. Section 9 adds this assumption and derives a further set of laws for sequential composition and iterations.

Section 12 builds on the general theory to provide a rely quotient operator that can be used to provide a general rely/guarantee framework for reasoning about concurrent programs.

2 Refinement Lattice

theory *Refinement-Lattice*

imports

Main

begin

unbundle *lattice-syntax*

The underlying lattice of commands is complete and distributive. We follow the refinement calculus tradition so that \sqcap is non-deterministic choice and $c \sqsubseteq d$ means c is refined (or implemented) by d .

declare *[[show-sorts]]*

Remove existing notation for quotient as it interferes with the rely quotient

no-notation *Equiv-Relations.quotient* (**infixl** $\langle'/'\rangle$ 90)

class *refinement-lattice* = *complete-distrib-lattice*

begin

The refinement lattice infimum corresponds to non-deterministic choice for commands.

abbreviation

refine :: 'a ⇒ 'a ⇒ bool (**infix** <⊑> 50)

where

$c \sqsubseteq d \equiv \text{less-eq } c \ d$

abbreviation

refine-strict :: 'a ⇒ 'a ⇒ bool (**infix** <⊒> 50)

where

$c \sqsupseteq d \equiv \text{less } c \ d$

Non-deterministic choice is monotonic in both arguments

lemma *inf-mono-left*: $a \sqsubseteq b \implies a \sqcap c \sqsubseteq b \sqcap c$

<proof>

lemma *inf-mono-right*: $c \sqsubseteq d \implies a \sqcap c \sqsubseteq a \sqcap d$

<proof>

Binary choice is a special case of choice over a set.

lemma *Inf2-inf*: $\sqcap \{f x \mid x. x \in \{c, d\}\} = f c \sqcap f d$

<proof>

Helper lemma for choice over indexed set.

lemma *INF-Inf*: $(\sqcap_{x \in X}. f x) = (\sqcap \{f x \mid x. x \in X\})$

<proof>

lemma (**in** -) *INF-absorb-args*: $(\sqcap i j. (f :: \text{nat} \Rightarrow 'c :: \text{complete-lattice}) (i + j)) = (\sqcap k. f k)$

<proof>

lemma (**in** -) *nested-Collect*: $\{f y \mid y. y \in \{g x \mid x. x \in X\}\} = \{f (g x) \mid x. x \in X\}$

<proof>

A transition lemma for INF distributivity properties, going from Inf to INF, qualified version followed by a straightforward one.

lemma *Inf-distrib-INF-qual*:

fixes *f* :: 'a ⇒ 'a ⇒ 'a

assumes *qual*: $P \{d x \mid x. x \in X\}$

assumes *f-Inf-distrib*: $\bigwedge c D. P D \implies f c (\sqcap D) = \sqcap \{f c d \mid d. d \in D\}$

shows $f c (\bigsqcap_{x \in X}. d x) = (\bigsqcap_{x \in X}. f c (d x))$
 $\langle proof \rangle$

lemma *Inf-distrib-INF*:

fixes $f :: 'a \Rightarrow 'a \Rightarrow 'a$

assumes f -*Inf-distrib*: $\bigwedge c D. f c (\bigsqcap D) = \bigsqcap \{f c d \mid d. d \in D\}$

shows $f c (\bigsqcap_{x \in X}. d x) = (\bigsqcap_{x \in X}. f c (d x))$

$\langle proof \rangle$

end

lemmas *refine-trans = order.trans*

More transitivity rules to make calculational reasoning smoother

declare *ord-eq-le-trans*[*trans*]

declare *ord-le-eq-trans*[*trans*]

declare *dual-order.trans*[*trans*]

abbreviation

dist-over-sup :: $('a::refinement-lattice \Rightarrow 'a) \Rightarrow bool$

where

dist-over-sup $F \equiv (\forall X. F (\bigsqcup X) = (\bigsqcup_{x \in X}. F (x)))$

abbreviation

dist-over-inf :: $('a::refinement-lattice \Rightarrow 'a) \Rightarrow bool$

where

dist-over-inf $F \equiv (\forall X. F (\bigsqcap X) = (\bigsqcap_{x \in X}. F (x)))$

end

3 Sequential Operator

theory *Sequential*

imports *Refinement-Lattice*

begin

3.1 Basic sequential

The sequential composition operator “;” is associative and has identity nil but it is not commutative. It has \perp as a left annihilator.

```

locale seq =
  fixes seq :: 'a::refinement-lattice  $\Rightarrow$  'a  $\Rightarrow$  'a (infixl <> 90)
  assumes seq-bot [simp]:  $\perp ; c = \perp$ 

```

```

locale nil =
  fixes nil :: 'a::refinement-lattice (<nil>)

```

The monoid axioms imply “;” is associative and has identity nil. Abort is a left annihilator of sequential composition.

```

locale sequential = seq + nil + seq: monoid seq nil
begin

```

```

declare seq.assoc [algebra-simps, field-simps]

```

```

lemmas seq-assoc = seq.assoc
lemmas seq-nil-right = seq.right-neutral
lemmas seq-nil-left = seq.left-neutral

```

```

end

```

3.2 Distributed sequential

Sequential composition distributes across arbitrary infima from the right but only across the binary (finite) infima from the left and hence it is monotonic in both arguments. We consider left distribution first. Note that Section 9 considers the case in which the weak-seq-inf-distrib axiom is strengthened to an equality.

```

locale seq-distrib-left = sequential +
  assumes weak-seq-inf-distrib:
    (c::'a::refinement-lattice);(d0  $\sqcap$  d1)  $\sqsubseteq$  (c;d0  $\sqcap$  c;d1)
begin

```

Left distribution implies sequential composition is monotonic in its right argument

```

lemma seq-mono-right: c0  $\sqsubseteq$  c1  $\Longrightarrow$  d ; c0  $\sqsubseteq$  d ; c1
  <proof>

```

```

lemma seq-bot-right [simp]: c ;  $\perp$   $\sqsubseteq$  c
  <proof>

```

```

end

```

```

locale seq-distrib-right = sequential +

```

assumes *Inf-seq-distrib*:

$(\sqcap C) ; d = (\sqcap (c :: 'a :: \text{refinement-lattice}) \in C. c ; d)$

begin

lemma *INF-seq-distrib*: $(\sqcap c \in C. f c) ; d = (\sqcap c \in C. f c ; d)$

<proof>

lemma *inf-seq-distrib*: $(c_0 \sqcap c_1) ; d = (c_0 ; d \sqcap c_1 ; d)$

<proof>

lemma *seq-mono-left*: $c_0 \sqsubseteq c_1 \implies c_0 ; d \sqsubseteq c_1 ; d$

<proof>

lemma *seq-top [simp]*: $\top ; c = \top$

<proof>

primrec *seq-power* :: $'a \Rightarrow \text{nat} \Rightarrow 'a$ (**infix** $\langle \cdot \rangle$ 80) **where**

seq-power-0: $a \langle \cdot \rangle 0 = \text{nil}$

| *seq-power-Suc*: $a \langle \cdot \rangle \text{Suc } n = a ; (a \langle \cdot \rangle n)$

notation (*latex output*)

seq-power $\langle \cdot \rangle$ [1000] 1000

notation (*HTML output*)

seq-power $\langle \cdot \rangle$ [1000] 1000

lemma *seq-power-front*: $(a \langle \cdot \rangle n) ; a = a ; (a \langle \cdot \rangle n)$

<proof>

lemma *seq-power-split-less*: $i < j \implies (b \langle \cdot \rangle j) = (b \langle \cdot \rangle i) ; (b \langle \cdot \rangle (j - i))$

<proof>

end

locale *seq-distrib* = *seq-distrib-right* + *seq-distrib-left*

begin

lemma *seq-mono*: $c_1 \sqsubseteq d_1 \implies c_2 \sqsubseteq d_2 \implies c_1 ; c_2 \sqsubseteq d_1 ; d_2$

<proof>

end

end

4 Parallel Operator

```
theory Parallel
imports Refinement-Lattice
begin
```

4.1 Basic parallel operator

The parallel operator is associative, commutative and has unit skip and has as an annihilator the lattice bottom.

```
locale skip =
  fixes skip :: 'a::refinement-lattice (⟨skip⟩)

locale par =
  fixes par :: 'a::refinement-lattice ⇒ 'a ⇒ 'a (infixl ⟨||⟩ 75)
  assumes abort-par: ⊥ || c = ⊥

locale parallel = par + skip + par: comm-monoid par skip
begin

lemmas [algebra-simps, field-simps] =
  par.assoc
  par commute
  par.left-commute

lemmas par-assoc = par.assoc
lemmas par-commute = par commute
lemmas par-skip = par.right-neutral
lemmas par-skip-left = par.left-neutral
```

end

4.2 Distributed parallel

The parallel operator distributes across arbitrary non-empty infima.

```
locale par-distrib = parallel +
  assumes par-Inf-distrib:  $D \neq \{\}$   $\implies c \parallel (\bigcap D) = (\bigcap_{d \in D} c \parallel d)$ 
```

```
begin
```

lemma *Inf-par-distrib*: $D \neq \{\}$ $\implies (\prod D) \parallel c = (\prod_{d \in D} d \parallel c)$
 ⟨proof⟩

lemma *par-INF-distrib*: $X \neq \{\}$ $\implies c \parallel (\prod_{x \in X} d x) = (\prod_{x \in X} c \parallel d x)$
 ⟨proof⟩

lemma *INF-par-distrib*: $X \neq \{\}$ $\implies (\prod_{x \in X} d x) \parallel c = (\prod_{x \in X} d x \parallel c)$
 ⟨proof⟩

lemma *INF-INF-par-distrib*:
 $X \neq \{\} \implies Y \neq \{\} \implies (\prod_{x \in X} c x) \parallel (\prod_{y \in Y} d y) = (\prod_{x \in X} \prod_{y \in Y} c x \parallel d y)$
 ⟨proof⟩

lemma *inf-par-distrib*: $(c_0 \sqcap c_1) \parallel d = (c_0 \parallel d) \sqcap (c_1 \parallel d)$
 ⟨proof⟩

lemma *inf-par-distrib2*: $d \parallel (c_0 \sqcap c_1) = (d \parallel c_0) \sqcap (d \parallel c_1)$
 ⟨proof⟩

lemma *inf-par-product*: $(a \sqcap b) \parallel (c \sqcap d) = (a \parallel c) \sqcap (a \parallel d) \sqcap (b \parallel c) \sqcap (b \parallel d)$
 ⟨proof⟩

lemma *par-mono*: $c_1 \sqsubseteq d_1 \implies c_2 \sqsubseteq d_2 \implies c_1 \parallel c_2 \sqsubseteq d_1 \parallel d_2$
 ⟨proof⟩

end
end

5 Weak Conjunction Operator

theory *Conjunction*
imports *Refinement-Lattice*
begin

The weak conjunction operator \mathfrak{m} is similar to least upper bound (\sqcup) but is abort strict, i.e. the lattice bottom is an annihilator: $c \mathfrak{m} \perp = \perp$. It has identity the command chaos that allows any non-aborting behaviour.

locale *chaos* =
fixes *chaos* :: 'a::refinement-lattice (⟨chaos⟩)

locale *conj* =
fixes *conj* :: 'a::refinement-lattice \Rightarrow 'a \Rightarrow 'a (**infixl** $\langle \mathfrak{m} \rangle$ 80)

assumes *conj-bot-right*: $c \pitchfork \perp = \perp$

Conjunction forms an idempotent, commutative monoid (i.e. a semi-lattice), with identity chaos.

locale *conjunction* = *conj* + *chaos* + *conj*: *semilattice-neutr conj chaos*

begin

lemmas [*algebra-simps*, *field-simps*] =
conj.assoc
conj.commute
conj.left-commute

lemmas *conj-assoc* = *conj.assoc*

lemmas *conj-commute* = *conj.commute*

lemmas *conj-idem* = *conj.idem*

lemmas *conj-chaos* = *conj.right-neutral*

lemmas *conj-chaos-left* = *conj.left-neutral*

lemma *conj-bot-left* [*simp*]: $\perp \pitchfork c = \perp$
<proof>

lemma *conj-not-bot*: $a \pitchfork b \neq \perp \implies a \neq \perp \wedge b \neq \perp$
<proof>

lemma *conj-distrib1*: $c \pitchfork (d_0 \pitchfork d_1) = (c \pitchfork d_0) \pitchfork (c \pitchfork d_1)$
<proof>

end

5.1 Distributed weak conjunction

The weak conjunction operator distributes across arbitrary non-empty infima.

locale *conj-distrib* = *conjunction* +

assumes *Inf-conj-distrib*: $D \neq \{\}$ $\implies (\bigcap D) \pitchfork c = (\bigcap_{d \in D} d \pitchfork c)$

begin

lemma *conj-Inf-distrib*: $D \neq \{\}$ $\implies c \pitchfork (\bigcap D) = (\bigcap_{d \in D} c \pitchfork d)$
<proof>

lemma *inf-conj-distrib*: $(c_0 \sqcap c_1) \pitchfork d = (c_0 \pitchfork d) \sqcap (c_1 \pitchfork d)$
<proof>

lemma *inf-conj-product*: $(a \sqcap b) \sqcap (c \sqcap d) = (a \sqcap c) \sqcap (a \sqcap d) \sqcap (b \sqcap c) \sqcap (b \sqcap d)$
<proof>

lemma *conj-mono*: $c_0 \sqsubseteq d_0 \implies c_1 \sqsubseteq d_1 \implies c_0 \sqcap c_1 \sqsubseteq d_0 \sqcap d_1$
<proof>

lemma *conj-mono-left*: $c_0 \sqsubseteq c_1 \implies c_0 \sqcap d \sqsubseteq c_1 \sqcap d$
<proof>

lemma *conj-mono-right*: $c_0 \sqsubseteq c_1 \implies d \sqcap c_0 \sqsubseteq d \sqcap c_1$
<proof>

lemma *conj-refine*: $c_0 \sqsubseteq d \implies c_1 \sqsubseteq d \implies c_0 \sqcap c_1 \sqsubseteq d$
<proof>

lemma *refine-to-conj*: $c \sqsubseteq d_0 \implies c \sqsubseteq d_1 \implies c \sqsubseteq d_0 \sqcap d_1$
<proof>

lemma *conjoin-non-aborting*: $\text{chaos} \sqsubseteq c \implies d \sqsubseteq d \sqcap c$
<proof>

lemma *conjunction-sup*: $c \sqcap d \sqsubseteq c \sqcup d$
<proof>

lemma *conjunction-sup-nonaborting*:
 assumes $\text{chaos} \sqsubseteq c$ **and** $\text{chaos} \sqsubseteq d$
 shows $c \sqcap d = c \sqcup d$
<proof>

lemma *conjoin-top*: $\text{chaos} \sqsubseteq c \implies c \sqcap \top = \top$
<proof>

end

end

6 Concurrent Refinement Algebra

This theory brings together the three main operators: sequential composition, parallel composition and conjunction, as well as the iteration operators.

theory *CRA*
imports

*Sequential
Conjunction
Parallel*

begin

Locale sequential-parallel brings together the sequential and parallel operators and relates their identities.

locale *sequential-parallel* = *seq-distrib* + *par-distrib* +
assumes *nil-par-nil*: $nil \parallel nil \sqsubseteq nil$
and *skip-nil*: $skip \sqsubseteq nil$
and *skip-skip*: $skip \sqsubseteq skip;skip$

begin

lemma *nil-absorb*: $nil \parallel nil = nil$ $\langle proof \rangle$

lemma *skip-absorb* [*simp*]: $skip;skip = skip$
 $\langle proof \rangle$

end

Locale conjunction-parallel brings together the weak conjunction and parallel operators and relates their identities. It also introduces the interchange axiom for conjunction and parallel.

locale *conjunction-parallel* = *conj-distrib* + *par-distrib* +
assumes *chaos-par-top*: $\top \sqsubseteq chaos \parallel \top$
assumes *chaos-par-chaos*: $chaos \sqsubseteq chaos \parallel chaos$
assumes *parallel-interchange*: $(c_0 \parallel c_1) \wp (d_0 \parallel d_1) \sqsubseteq (c_0 \wp d_0) \parallel (c_1 \wp d_1)$

begin

lemma *chaos-skip*: $chaos \sqsubseteq skip$
 $\langle proof \rangle$

lemma *chaos-par-chaos-eq*: $chaos = chaos \parallel chaos$
 $\langle proof \rangle$

lemma *nonabort-par-top*: $chaos \sqsubseteq c \implies c \parallel \top = \top$
 $\langle proof \rangle$

lemma *skip-conj-top*: $skip \wp \top = \top$
 $\langle proof \rangle$

lemma *conj-distrib2*: $c \sqsubseteq c \parallel c \implies c \wp (d_0 \parallel d_1) \sqsubseteq (c \wp d_0) \parallel (c \wp d_1)$
 $\langle proof \rangle$

end

Locale conjunction-sequential brings together the weak conjunction and sequential operators. It also introduces the interchange axiom for conjunction and sequential.

locale *conjunction-sequential* = *conj-distrib* + *seq-distrib* +
 assumes *chaos-seq-chaos*: $chaos \sqsubseteq chaos;chaos$
 assumes *sequential-interchange*: $(c_0;c_1) \pitchfork (d_0;d_1) \sqsubseteq (c_0 \pitchfork d_0);(c_1 \pitchfork d_1)$
begin

lemma *chaos-nil*: $chaos \sqsubseteq nil$
 $\langle proof \rangle$

lemma *chaos-seq-absorb*: $chaos = chaos;chaos$
 $\langle proof \rangle$

lemma *seq-bot-conj*: $c;\perp \pitchfork d \sqsubseteq (c \pitchfork d);\perp$
 $\langle proof \rangle$

lemma *conj-seq-bot-right* [*simp*]: $c;\perp \pitchfork c = c;\perp$
 $\langle proof \rangle$

lemma *conj-distrib3*: $c \sqsubseteq c;c \implies c \pitchfork (d_0 ; d_1) \sqsubseteq (c \pitchfork d_0);(c \pitchfork d_1)$
 $\langle proof \rangle$

end

Locale *cra* brings together sequential, parallel and weak conjunction.

locale *cra* = *sequential-parallel* + *conjunction-parallel* + *conjunction-sequential*

end

7 Galois Connections and Fusion Theorems

theory *Galois-Connections*
imports *Refinement-Lattice*
begin

The concept of Galois connections is introduced here to prove the fixed-point fusion lemmas. The definition of Galois connections used is quite simple but

encodes a lot of information. The material in this section is largely based on the work of the Eindhoven Mathematics of Program Construction Group [1] and the reader is referred to their work for a full explanation of this section.

7.1 Lower Galois connections

lemma *Collect-2set* [*simp*]: $\{F x \mid x. x = a \vee x = b\} = \{F a, F b\}$
 ⟨*proof*⟩

locale *lower-galois-connections*
begin

definition

l-adjoint :: (*'a*::*refinement-lattice* \Rightarrow *'a*) \Rightarrow (*'a* \Rightarrow *'a*) ($\langle \cdot^b \rangle$ [201] 200)

where

$(F^b) x \equiv \sqcap \{y. x \sqsubseteq F y\}$

lemma *dist-inf-mono*:

assumes *distF*: *dist-over-inf F*

shows *mono F*

⟨*proof*⟩

lemma *l-cancellation*: *dist-over-inf F* $\Longrightarrow x \sqsubseteq (F \circ F^b) x$

⟨*proof*⟩

lemma *l-galois-connection*: *dist-over-inf F* $\Longrightarrow ((F^b) x \sqsubseteq y) \longleftrightarrow (x \sqsubseteq F y)$

⟨*proof*⟩

lemma *v-simple-fusion*: *mono G* $\Longrightarrow \forall x. ((F \circ G) x \sqsubseteq (H \circ F) x) \Longrightarrow F (\text{gfp } G) \sqsubseteq \text{gfp } H$

⟨*proof*⟩

7.2 Greatest fixpoint fusion theorems

Combining lower Galois connections and greatest fixed points allows elegant proofs of the weak fusion lemmas.

theorem *fusion-gfp-geq*:

assumes *monoH*: *mono H*

and *distribF*: *dist-over-inf F*

and *comp-geq*: $\bigwedge x. ((H \circ F) x \sqsubseteq (F \circ G) x)$

shows *gfp H* $\sqsubseteq F (\text{gfp } G)$

⟨*proof*⟩

theorem *fusion-gfp-eq*:
assumes *monoH*: *mono H* **and** *monoG*: *mono G*
and *distF*: *dist-over-inf F*
and *fgh-comp*: $\bigwedge x. ((F \circ G) x = (H \circ F) x)$
shows $F (\text{gfp } G) = \text{gfp } H$
 $\langle \text{proof} \rangle$

end

7.3 Upper Galois connections

locale *upper-galois-connections*
begin

definition

u-adjoint :: (*a*::*refinement-lattice* \Rightarrow *a*) \Rightarrow (*a* \Rightarrow *a*) ($\leftarrow \#$) [201] 200)

where

$(F\#) x \equiv \bigsqcup \{y. F y \sqsubseteq x\}$

lemma *dist-sup-mono*:

assumes *distF*: *dist-over-sup F*

shows *mono F*

$\langle \text{proof} \rangle$

lemma *u-cancellation*: *dist-over-sup F* \Longrightarrow $(F \circ F\#) x \sqsubseteq x$

$\langle \text{proof} \rangle$

lemma *u-galois-connection*: *dist-over-sup F* \Longrightarrow $(F x \sqsubseteq y) \longleftrightarrow (x \sqsubseteq (F\#) y)$

$\langle \text{proof} \rangle$

lemma *u-simple-fusion*: *mono H* \Longrightarrow $\forall x. ((F \circ G) x \sqsubseteq (G \circ H) x) \Longrightarrow \text{lfp } F \sqsubseteq G (\text{lfp } H)$

$\langle \text{proof} \rangle$

7.4 Least fixpoint fusion theorems

Combining upper Galois connections and least fixed points allows elegant proofs of the strong fusion lemmas.

theorem *fusion-lfp-leq*:

assumes *monoH*: *mono H*

and *distribF*: *dist-over-sup F*

and *comp-leq*: $\bigwedge x. ((F \circ G) x \sqsubseteq (H \circ F) x)$

shows $F (\text{lfp } G) \sqsubseteq (\text{lfp } H)$
<proof>

theorem *fusion-lfp-eq*:
assumes *monoH*: *mono H* **and** *monoG*: *mono G*
and *distF*: *dist-over-sup F*
and *fgh-comp*: $\bigwedge x. ((F \circ G) x = (H \circ F) x)$
shows $F (\text{lfp } G) = (\text{lfp } H)$
<proof>

end
end

8 Iteration

theory *Iteration*
imports
 Galois-Connections
 CRA
begin

8.1 Possibly infinite iteration

Iteration of finite or infinite steps can be defined using a least fixed point.

locale *finite-or-infinite-iteration* = *seq-distrib* + *upper-galois-connections*
begin

definition
iter :: 'a \Rightarrow 'a *<math>\langle \leftarrow^\omega \rangle [103] 102)*

where
 $c^\omega \equiv \text{lfp } (\lambda x. \text{nil} \sqcap c;x)$

lemma *iter-step-mono*: *mono* $(\lambda x. \text{nil} \sqcap c;x)$
<proof>

This fixed point definition leads to the two core iteration lemmas: folding and induction.

theorem *iter-unfold*: $c^\omega = \text{nil} \sqcap c;c^\omega$
<proof>

lemma iter-induct-nil: $nil \sqcap c; x \sqsubseteq x \implies c^\omega \sqsubseteq x$
<proof>

lemma iter0: $c^\omega \sqsubseteq nil$
<proof>

lemma iter1: $c^\omega \sqsubseteq c$
<proof>

lemma iter2 [simp]: $c^\omega; c^\omega = c^\omega$
<proof>

lemma iter-mono: $c \sqsubseteq d \implies c^\omega \sqsubseteq d^\omega$
<proof>

lemma iter-abort: $\perp = nil^\omega$
<proof>

lemma nil-iter: $\top^\omega = nil$
<proof>

end

8.2 Finite iteration

Iteration of a finite number of steps (Kleene star) is defined using the greatest fixed point.

locale *finite-iteration* = *seq-distrib* + *lower-galois-connections*
begin

definition

fiter :: 'a \Rightarrow 'a ($\langle \cdot^* \rangle$ [101] 100)

where

$c^* \equiv \text{gfp } (\lambda x. nil \sqcap c; x)$

lemma fin-iter-step-mono: *mono* $(\lambda x. nil \sqcap c; x)$
<proof>

This definition leads to the two core iteration lemmas: folding and induction.

lemma fiter-unfold: $c^* = nil \sqcap c; c^*$
<proof>

lemma *fiter-induct-nil*: $x \sqsubseteq \text{nil} \sqcap c; x \implies x \sqsubseteq c^*$
<proof>

lemma *fiter0*: $c^* \sqsubseteq \text{nil}$
<proof>

lemma *fiter1*: $c^* \sqsubseteq c$
<proof>

lemma *fiter-induct-eq*: $c^*; d = \text{gfp } (\lambda x. c; x \sqcap d)$
<proof>

theorem *fiter-induct*: $x \sqsubseteq d \sqcap c; x \implies x \sqsubseteq c^*; d$
<proof>

lemma *fiter2* [*simp*]: $c^*; c^* = c^*$
<proof>

lemma *fiter3* [*simp*]: $(c^*)^* = c^*$
<proof>

lemma *fiter-mono*: $c \sqsubseteq d \implies c^* \sqsubseteq d^*$
<proof>

end

8.3 Infinite iteration

Iteration of infinite number of steps can be defined using a least fixed point.

locale *infinite-iteration* = *seq-distrib* + *lower-galois-connections*
begin

definition

infiter :: 'a \Rightarrow 'a ($\langle -^\infty \rangle$ [105] 106)

where

$c^\infty \equiv \text{lfp } (\lambda x. c; x)$

lemma *infiter-step-mono*: *mono* $(\lambda x. c; x)$
<proof>

This definition leads to the two core iteration lemmas: folding and induction.

theorem *infiter-unfold*: $c^\infty = c; c^\infty$
<proof>

lemma *infiter-induct*: $c;x \sqsubseteq x \implies c^\infty \sqsubseteq x$
<proof>

theorem *infiter-unfold-any*: $c^\infty = (c \ ;^i \ i) ; c^\infty$
<proof>

lemma *infiter-annil*: $c^\infty ; x = c^\infty$
<proof>

end

8.4 Combined iteration

The three different iteration operators can be combined to show that finite iteration refines finite-or-infinite iteration.

locale *iteration = finite-or-infinite-iteration + finite-iteration + infinite-iteration*

begin

lemma *refine-iter*: $c^\omega \sqsubseteq c^*$
<proof>

lemma *iter-absorption [simp]*: $(c^\omega)^* = c^\omega$
<proof>

lemma *infiter-inf-top*: $c^\infty = c^\omega ; \top$
<proof>

lemma *infiter-fiter-top*:
shows $c^\infty \sqsubseteq c^* ; \top$
<proof>

lemma *inf-ref-infiter*: $c^\omega \sqsubseteq c^\infty$
<proof>

end

end

9 Sequential composition for conjunctive models

theory *Conjunctive-Sequential*

imports *Sequential*

begin

Sequential left-distributivity is only supported by conjunctive models but does not apply in general. The relational model is one such example.

locale *seq-finite-conjunctive* = *seq-distrib-right* +
assumes *seq-inf-distrib*: $c; (d_0 \sqcap d_1) = c; d_0 \sqcap c; d_1$

begin

sublocale *seq-distrib-left*

<proof>

end

locale *seq-infinite-conjunctive* = *seq-distrib-right* +
assumes *seq-Inf-distrib*: $D \neq \{\}$ $\implies c; \sqcap D = (\sqcap d \in D. c; d)$

begin

sublocale *seq-distrib*

<proof>

lemma *seq-INF-distrib*: $X \neq \{\} \implies c; (\sqcap x \in X. d x) = (\sqcap x \in X. c; d x)$

<proof>

lemma *seq-INF-distrib-UNIV*: $c; (\sqcap x. d x) = (\sqcap x. c; d x)$

<proof>

lemma *INF-INF-*seq-distrib**: $Y \neq \{\} \implies (\sqcap x \in X. c x); (\sqcap y \in Y. d y) = (\sqcap x \in X. \sqcap y \in Y. c x; d y)$

<proof>

lemma *INF-INF-*seq-distrib-UNIV**: $(\sqcap x. c x); (\sqcap y. d y) = (\sqcap x. \sqcap y. c x; d y)$

<proof>

end

end

10 Infimum nat lemmas

theory *Infimum-Nat*

imports

Refinement-Lattice

begin

locale *infimum-nat*

begin

lemma *INF-partition-nat3*:

fixes $f :: nat \Rightarrow nat \Rightarrow 'a::refinement-lattice$

shows $(\prod j. f i j) =$

$(\prod j \in \{j. i = j\}. f i j) \sqcap$

$(\prod j \in \{j. i < j\}. f i j) \sqcap$

$(\prod j \in \{j. j < i\}. f i j)$

$\langle proof \rangle$

lemma *INF-INF-partition-nat3*:

fixes $f :: nat \Rightarrow nat \Rightarrow 'a::refinement-lattice$

shows $(\prod i. \prod j. f i j) =$

$(\prod i. \prod j \in \{j. i = j\}. f i j) \sqcap$

$(\prod i. \prod j \in \{j. i < j\}. f i j) \sqcap$

$(\prod i. \prod j \in \{j. j < i\}. f i j)$

$\langle proof \rangle$

lemma *INF-nat-shift*: $(\prod i \in \{i. 0 < i\}. f i) = (\prod i. f (Suc i))$

$\langle proof \rangle$

lemma *INF-nat-minus*:

fixes $f :: nat \Rightarrow 'a::refinement-lattice$

shows $(\prod j \in \{j. i < j\}. f (j - i)) = (\prod k \in \{k. 0 < k\}. f k)$

$\langle proof \rangle$

lemma *INF-INF-guarded-switch*:

fixes $f :: nat \Rightarrow nat \Rightarrow 'a::refinement-lattice$

shows $(\prod i. \prod j \in \{j. j < i\}. f j (i - j)) = (\prod j. \prod i \in \{i. j < i\}. f j (i - j))$

$\langle proof \rangle$

end

end

11 Iteration for conjunctive models

theory *Conjunctive-Iteration*

imports

Conjunctive-Sequential

Iteration

Infimum-Nat

begin

Sequential left-distributivity is only supported by conjunctive models but does not apply in general. The relational model is one such example.

locale *iteration-finite-conjunctive* = *seq-finite-conjunctive* + *iteration*

begin

lemma *isolation*: $c^\omega = c^* \sqcap c^\infty$

<proof>

lemma *iter-induct-isolate*: $c^*;d \sqcap c^\infty = \text{lfp } (\lambda x. d \sqcap c;x)$

<proof>

lemma *iter-induct-eq*: $c^\omega;d = \text{lfp } (\lambda x. d \sqcap c;x)$

<proof>

lemma *iter-induct*: $d \sqcap c;x \sqsubseteq x \implies c^\omega;d \sqsubseteq x$

<proof>

lemma *iter-isolate*: $c^*;d \sqcap c^\infty = c^\omega;d$

<proof>

lemma *iter-isolate2*: $c;c^*;d \sqcap c^\infty = c;c^\omega;d$

<proof>

lemma *iter-decomp*: $(c \sqcap d)^\omega = c^\omega;(d;c^\omega)^\omega$

<proof>

lemma *iter-leapfrog-var*: $(c;d)^\omega;c \sqsubseteq c;(d;c)^\omega$

<proof>

lemma *iter-leapfrog*: $c;(d;c)^\omega = (c;d)^\omega;c$

<proof>

lemma *fiter-leapfrog*: $c;(d;c)^* = (c;d)^*;c$

<proof>

end

locale *iteration-infinite-conjunctive* = *seq-infinite-conjunctive* + *iteration* + *inifimum-nat*

begin

lemma *fiter-seq-choice*: $c^* = (\prod i::nat. c \ ;^i i)$

<proof>

lemma *fiter-seq-choice-nonempty*: $c \ ; c^* = (\prod i \in \{i. 0 < i\}. c \ ;^i i)$

<proof>

end

locale *conj-iteration* = *cra* + *iteration-infinite-conjunctive*

begin

lemma *conj-distrib4*: $c^* \ \cap \ d^* \sqsubseteq (c \ \cap \ d)^*$

<proof>

end

end

12 Rely Quotient Operator

The rely quotient operator is used to generalise a Jones-style rely condition to a process [5]. It is defined in terms of the parallel operator and a process i representing interference from the environment.

theory *Rely-Quotient*

imports

CRA

Conjunctive-Iteration

begin

12.1 Basic rely quotient

The rely quotient of a process c and an interference process i is the most general process d such that c is refined by $d \parallel i$. The following locale introduces the definition of the rely quotient $c // i$ as a non-deterministic choice over all processes d such that c is refined by $d \parallel i$.

locale *rely-quotient* = *par-distrib* + *conjunction-parallel*
begin

definition

rely-quotient :: 'a \Rightarrow 'a \Rightarrow 'a (**infixl** <'//> 85)

where

$c // i \equiv \sqcap \{ d. (c \sqsubseteq d \parallel i) \}$

Any process c is implemented by itself if the interference is skip.

lemma *quotient-identity*: $c // \text{skip} = c$
 <proof>

Provided the interference process i is non-aborting (i.e. it refines chaos), any process c is refined by its rely quotient with i in parallel with i . If interference i was allowed to be aborting then, because $(c // \perp) \parallel \perp$ equals \perp , it does not refine c in general.

theorem *rely-quotient*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$

shows $c \sqsubseteq (c // i) \parallel i$

<proof>

The following theorem represents the Galois connection between the parallel operator (upper adjoint) and the rely quotient operator (lower adjoint). This basic relationship is used to prove the majority of the theorems about rely quotient.

theorem *rely-refinement*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$

shows $c // i \sqsubseteq d \iff c \sqsubseteq d \parallel i$

<proof>

Refining the “numerator” in a quotient, refines the quotient.

lemma *rely-mono*:

assumes *c-refsto-d*: $c \sqsubseteq d$

shows $(c // i) \sqsubseteq (d // i)$

<proof>

Refining the “denominator” in a quotient, gives a reverse refinement for the quotients. This corresponds to weaken rely condition law of Jones [5], i.e. assuming

less about the environment.

lemma *weaken-rely*:

assumes *i-refsto-j*: $i \sqsubseteq j$

shows $(c // j) \sqsubseteq (c // i)$

<proof>

lemma *par-nonabort*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$

assumes *nonabort-j*: $\text{chaos} \sqsubseteq j$

shows $\text{chaos} \sqsubseteq i \parallel j$

<proof>

Nesting rely quotients of j and i means the same as a single quotient which is the parallel composition of i and j .

lemma *nested-rely*:

assumes *j-nonabort*: $\text{chaos} \sqsubseteq j$

shows $((c // j) // i) = c // (i \parallel j)$

<proof>

end

12.2 Distributed rely quotient

locale *rely-distrib* = *rely-quotient* + *conjunction-sequential*

begin

The following is a fundamental law for introducing a parallel composition of process to refine a conjunction of specifications. It represents an abstract view of the parallel introduction law of Jones [5].

lemma *introduce-parallel*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$

assumes *nonabort-j*: $\text{chaos} \sqsubseteq j$

shows $c \text{ m } d \sqsubseteq (j \text{ m } (c // i)) \parallel (i \text{ m } (d // j))$

<proof>

Rely quotients satisfy a range of distribution properties with respect to the other operators.

lemma *distribute-rely-conjunction*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$

shows $(c \text{ m } d) // i \sqsubseteq (c // i) \text{ m } (d // i)$

<proof>

lemma *distribute-rely-choice*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
shows $(c \sqcap d) // i \sqsubseteq (c // i) \sqcap (d // i)$
<proof>

lemma *distribute-rely-parallel1*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes *nonabort-j*: $\text{chaos} \sqsubseteq j$
shows $(c \parallel d) // (i \parallel j) \sqsubseteq (c // i) \parallel (d // j)$
<proof>

lemma *distribute-rely-parallel2*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes *i-par-i*: $i \parallel i \sqsubseteq i$
shows $(c \parallel d) // i \sqsubseteq (c // i) \parallel (d // i)$
<proof>

lemma *distribute-rely-sequential*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes $(\forall c. (\forall d. ((c \parallel i);(d \parallel i) \sqsubseteq (c;d) \parallel i)))$
shows $(c;d) // i \sqsubseteq (c // i);(d // i)$
<proof>

lemma *distribute-rely-sequential-event*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes *nonabort-j*: $\text{chaos} \sqsubseteq j$
assumes *nonabort-e*: $\text{chaos} \sqsubseteq e$
assumes $(\forall c. (\forall d. ((c \parallel i);e;(d \parallel j) \sqsubseteq (c;e;d) \parallel (i;e;j))))$
shows $(c;e;d) // (i;e;j) \sqsubseteq (c // i);e;(d // j)$
<proof>

lemma *introduce-parallel-with-rely*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes *nonabort-j0*: $\text{chaos} \sqsubseteq j_0$
assumes *nonabort-j1*: $\text{chaos} \sqsubseteq j_1$
shows $(c \sqcap d) // i \sqsubseteq (j_1 \sqcap (c // (j_0 \parallel i))) \parallel (j_0 \sqcap (d // (j_1 \parallel i)))$
<proof>

lemma *introduce-parallel-with-rely-guarantee*:

assumes *nonabort-i*: $\text{chaos} \sqsubseteq i$
assumes *nonabort-j0*: $\text{chaos} \sqsubseteq j_0$
assumes *nonabort-j1*: $\text{chaos} \sqsubseteq j_1$
shows $(j_1 \parallel j_0) \sqcap (c \sqcap d) // i \sqsubseteq (j_1 \sqcap (c // (j_0 \parallel i))) \parallel (j_0 \sqcap (d // (j_1 \parallel i)))$

$\langle proof \rangle$

lemma *wrap-rely-guar*:

assumes *nonabort-rg*: $chaos \sqsubseteq rg$

and *skippable*: $rg \sqsubseteq skip$

shows $c \sqsubseteq rg \sqcap c // rg$

$\langle proof \rangle$

end

locale *rely-distrib-iteration* = *rely-distrib* + *iteration-finite-conjunctive*

begin

lemma *distribute-rely-iteration*:

assumes *nonabort-i*: $chaos \sqsubseteq i$

assumes $(\forall c. (\forall d. ((c \parallel i); (d \parallel i) \sqsubseteq (c;d \parallel i)))$

shows $(c^\omega; d) // i \sqsubseteq (c // i)^\omega; (d // i)$

$\langle proof \rangle$

end

end

13 Conclusions

The theories presented here provide a quite abstract view of the rely/guarantee approach to concurrent program refinement. A trace semantics for this theory has been developed [2]. The concurrent refinement algebra is general enough to also form the basis of a more concrete rely/guarantee approach based on a theory of atomic steps and synchronous parallel and weak conjunction operators [4].

Acknowledgements. This research was supported by Australian Research Council Grant grant DP130102901 and EPSRC (UK) Taming Concurrency grant. This research has benefited from feedback from Robert Colvin, Chelsea Edmonds, Ned Hoy, Cliff Jones, Larissa Meinicke, and Kirsten Winter.

A Differences to earlier paper

This appendix summarises the differences between these Isabelle theories and the earlier paper [3]. We list the changes to the axioms but not all the flow on effects to lemmas.

1. The earlier paper assumes $c;(d_0 \sqcap d_1) = (c;d_0) \sqcap (c;d_1)$ but here we separate the case where this is only a refinement from left to right (Section 3) from the equality case (Section 9).
2. The earlier paper assumes $(\sqcap C) \parallel d = (\sqcap c \in C.c \parallel d)$ but in Section 4 we assume this only for non-empty C and furthermore assume that parallel is abort strict, i.e. $\perp \parallel c = c$.
3. The earlier paper assumes $c \pitchfork (\sqcup D) = (\sqcup d \in D.c \pitchfork d)$. In Section 5 that assumption is not made because it does not hold for the model we have in mind [2] but we do assume $c \pitchfork \perp = \perp$.
4. In Section 6 we add the assumption $nil \sqsubseteq nil \parallel nil$ to locale sequential-parallel.
5. In Section 6 we add the assumption $\top \sqsubseteq chaos \parallel \top$.
6. In Section 6 we assume only $chaos \sqsubseteq chaos \parallel chaos$ whereas in the paper this is an equality (the reverse direction is straightforward to prove).
7. In Section 6 axiom chaos-skip ($chaos \sqsubseteq skip$) has been dropped because it can be proven as a lemma using the parallel-interchange axiom.
8. In Section 6 we add the assumption $chaos \sqsubseteq chaos ; chaos$.
9. Section 9 assumes $D \neq \{\} \Rightarrow c ; \sqcap D = (\sqcap d \in D.c ; d)$. This distribution axiom is not considered in the earlier paper.
10. Because here parallel does not distribute over an empty non-deterministic choice (see point 2 above) in Section 12 the theorem rely-quotient needs to assume the interference process i is non-aborting (refines chaos). This also affects many lemmas in this section that depend on theorem rely-quotient.

References

- [1] C. Aarts, R. Backhouse, E. Boiten, H. Doombos, N. van Gasteren, R. van Geldrop, P. Hoogendijk, E. Voermans, and J. van der Woude. Fixed-point

- calculus. *Information Processing Letters*, 53:131–136, 1995. Mathematics of Program Construction Group.
- [2] R. J. Colvin, I. J. Hayes, and L. A. Meinicke. Designing a semantic model for a wide-spectrum language with concurrency. *Formal Aspects of Computing*, pages 1–22, 2016. Accepted 28 November 2016.
- [3] I. J. Hayes. Generalised rely-guarantee concurrency: An algebraic foundation. *Formal Aspects of Computing*, 28(6):1057–1078, November 2016.
- [4] I. J. Hayes, R. J. Colvin, L. A. Meinicke, K. Winter, and A. Velykis. An algebra of synchronous atomic steps. In J. Fitzgerald, C. Heitmeyer, S. Gnesi, and A. Philippou, editors, *FM 2016: Formal Methods: 21st International Symposium, Limassol, Cyprus, November 9-11, 2016, Proceedings*, volume 9995 of *Lecture Notes in Computer Science*, pages 352–369, Cham, November 2016. Springer International Publishing.
- [5] C. Jones. Tentative steps toward a development method for interfering programs. *ACM Transactions on Programming Languages and Systems*, 5(4):596–619, Oct. 1983.
- [6] C. B. Jones. *Development Methods for Computer Programs including a Notion of Interference*. PhD thesis, Oxford University, June 1981. Available as: Oxford University Computing Laboratory (now Computer Science) Technical Monograph PRG-25.